Kapitel 16: Fourier-Reihen und FFT

Bei die Analyse periodischer Vorgänge zerlegt man ein Signal in seine harmonischen Bestandteile. Hierzu verwendet man die Formeln für Fourier-Reihen. Die Fourier-Reihe ist eine Darstellung der Funktion f(t) als Superposition von Sinus- und Kosinusfunktionen mit den Fourier-Koeffizienten als zugehörigen Amplituden.

Im Abschnitt Fourier-Reihen (analytisch) werden unter Verwendung des **int**-Befehls die reellen Fourier-Koeffizienten formelmäßig berechnet. Hierbei darf die Funktion f(t) Parameter enthalten. Ist man nur an zahlenmäßigen Werten der Ergebnisse interessiert, so verwendet man die Befehlsfolge aus dem Abschnitt Fourier-Reihen (numerisch). Dann darf die Funktion keine Parameter enthalten.

Im Abschnitt über die komplexen Fourier-Reihen erfolgt eine Zerlegung des Signals f(t) in Anteile der komplexwertigen Exponentialfunktion $e^{i\omega_n t}$. Die zugehörigen komplexen Fourier-Koeffizienten c_n entsprechen bis auf den Faktor 2 dem Amplitudenspektrum der Funktion. Im Abschnitt FFT (**F**ast **F**ourier **T**ransformation) wird die Fourier-Analysis mit dem **FourierTransform**-Befehl für diskrete Werte durchgeführt. Die FFT wird auch für die Analyse von Messwerten verwendet, die in einem Zeitbereich [0, T] abgetastet vorliegen. Man beachte, dass der **FourierTransform**-Befehl aus dem **DiscreteTransforms**-Package den bisherigen **FFT**-Befehl ersetzt.

Bei der numerischen Berechnung der Fourier-Koeffizienten und bei der Anwendung der FFT darf die Funktion keine Parameter enthalten. Bei der numerischen Berechnung der Integrale können Koeffizienten, die analytisch zwar Null sind, nun Werte in der Größenordnung 10^{-9} und kleiner bekommen. Durch die Angabe **Digits**:=n wird dann die Darstellung der Zahlen und die Genauigkeit der Rechnung auf n Stellen erhöht. Standardmäßig wird mit 10 Stellen gerechnet.

16.1 Fourier-Reihen (analytisch)

	worksheet
Problem	Gegeben ist eine T-periodische Funktion f(t). Gesucht sind die Fourier-Koeffizienten a_0, a_n und b_n: $$a_0 = \frac{1}{T}\int_0^T f(t)\,dt,$$ $$a_n = \frac{2}{T}\int_0^T f(t)\cos(n\,w_0\,t)\,dt,$$ $$b_n = \frac{2}{T}\int_0^T f(t)\sin(n\,w_0\,t)\,dt \quad \text{mit} \quad w_0 = \frac{2\pi}{T}$$ sowie die Darstellung der Funktion über die Fourier-Reihe $$f(t) = a_0 + \left(\sum_{n=1}^{\infty} a_n \cos(n\,w_0\,t)\right) + \left(\sum_{n=1}^{\infty} b_n \sin(n\,w_0\,t)\right)$$
Befehl	Maple-Befehlsfolge
Parameter	-

Beispiel	Gesucht sind formelmäßige Ausdrücke für die Koeffizienten und die Reihendarstellung einer Dreiecksfunktion
	`> f1:=t:                #erstes Intervall 0<=t<=T/4`
	`> f2:=-1/3*(t-T):       #zweites Intervall`
	Gleichanteil der Funktion
	`> a[0]:=1/T*(int(f1,t=0..T/4)+int(f2,t=T/4..T));`
	$$a_0 := \frac{1}{8}T$$
	Koeffizienten a_n
	`> a[n]:=2/T*(int(f1*cos(n*2*Pi/T*t),t=0..T/4)`
	`        +int(f2*cos(n*2*Pi/T*t),t=T/4..T)):`
	`> a[n]:=normal(a[n]);`
	$$a_n := -\frac{1}{3}\,\frac{T\left(-2\cos\left(\frac{1}{2}n\pi\right)+1+\cos(n\pi)^2\right)}{n^2\pi^2}$$

Koeffizienten b_n

```
> b[n]:=2/T*(int(f1*sin(n*2*Pi/T*t),t=0..T/4)
          +int(f2*sin(n*2*Pi/T*t),t=T/4..T)):
> b[n]:=normal(b[n]);
```

$$b_n := -\frac{1}{3}\,\frac{T\left(-2\sin\left(\frac{1}{2}\,n\,\pi\right)+\sin(n\,\pi)\cos(n\,\pi)\right)}{n^2\,\pi^2}$$

Darstellung der Partialsumme für N=10

```
> N:=10: T:=1:
> a[0] + sum(a[n]*cos(n*2*Pi/T*t), n=1..N) +
       sum(b[n]*sin(n*2*Pi/T*t), n=1..N);
```

$$\frac{1}{8}-\frac{2}{3}\,\frac{\cos(2\,\pi\,t)}{\pi^2}-\frac{1}{3}\,\frac{\cos(4\,\pi\,t)}{\pi^2}-\frac{2}{27}\,\frac{\cos(6\,\pi\,t)}{\pi^2}-\frac{2}{75}\,\frac{\cos(10\,\pi\,t)}{\pi^2}$$
$$-\frac{1}{27}\,\frac{\cos(12\,\pi\,t)}{\pi^2}-\frac{2}{147}\,\frac{\cos(14\,\pi\,t)}{\pi^2}-\frac{2}{243}\,\frac{\cos(18\,\pi\,t)}{\pi^2}-\frac{1}{75}\,\frac{\cos(20\,\pi\,t)}{\pi^2}$$
$$\frac{\frac{2}{3}\sin(2\,\pi\,t)}{\pi^2}-\frac{2}{27}\,\frac{\sin(6\,\pi\,t)}{\pi^2}+\frac{\frac{2}{75}\sin(10\,\pi\,t)}{\pi^2}-\frac{2}{147}\,\frac{\sin(14\,\pi\,t)}{\pi^2}+\frac{\frac{2}{243}\sin(18\,\pi\,t}{\pi^2}$$

```
> plot(f_reihe, t=0..2*T, color=red);
```



Hinweise

Bei der analytischen Berechnung der Fourier-Koeffizienten dürfen in der Funktion Parameter enthalten sein. Damit das Integral aber berechnet wird, sollte auf eine Definition der Funktion über den **piecewise**-Befehl verzichtet werden. Stattdessen werden die Integrale geeignet aufgespaltet und die Funktionsvorschrift direkt in die Integrale eingesetzt.
Durch die Ergänzung *assuming* bei der Bestimmung der Koeffizienten `> a[n]:=normal(a[n]) assuming n::posint;` wird die Berechnung unter der Voraussetzung durchgeführt, dass n eine positive ganze Zahl darstellt. Dann werden Terme der Form sin(n π) automatisch zu Null vereinfacht.
Mit dem **plot**-Befehl wird die Partialsumme gezeichnet.

Siehe auch

`int`, `normal`, `plot`; → Fourier-Reihen (numerisch) → FFT.

16.2 Fourier-Reihen (numerisch)

worksheet

Problem

Gegeben ist eine T-periodische Funktion f(t). Gesucht sind die numerisch berechneten Fourier-Koeffizienten bis zur Ordnung N

$$a_0 = \frac{1}{T} \int_0^T \mathrm{f}(t)\, dt\,,$$

$$a_n = \frac{2}{T} \int_0^T \mathrm{f}(t) \cos(n\, w_0\, t)\, dt\,,$$

$$b_n = \frac{2}{T} \int_0^T \mathrm{f}(t) \sin(n\, w_0\, t)\, dt \quad \text{mit} \quad w_0 = \frac{2\pi}{T}$$

sowie die Partialsumme der Fourier-Reihe

$$\mathrm{f}(t) = a_0 + \left(\sum_{n=1}^{N} a_n \cos(n\, w_0\, t) \right) + \left(\sum_{n=1}^{N} b_n \sin(n\, w_0\, t) \right)$$

Befehl

Maple-Befehlsfolge

Parameter

-

Beispiel

Gesucht ist die Fourier-Reihe einer Dreiecksfunktion

```
> f:=piecewise(t<T/4,t, t<T,-1/3*(t-T),0):

> T:=10:
> a[0]:=1/T*Int(f,t=0..T):
> a[0]:=evalf(%);
```

$$a_0 := 1.250000000$$

```
> N:=10:
> for n from 1 to N
> do
>   a[n]:=2/T*Int(f*cos(n*2*Pi/T*t),t=0..T);
>   a[n]:=evalf(%);
>   b[n]:=2/T*Int(f*sin(n*2*Pi/T*t),t=0..T);
>   b[n]:=evalf(%);
>   printf(`n=%2d:  %+8.4e %+8.4e.\n`,
                              n, a[n], b[n]);
> end do:
    n= 1:  -6.7547e-01 +6.7547e-01.
    n= 2:  -3.3774e-01 +0.0000e-01.
    n= 3:  -7.5053e-02 -7.5053e-02.
    n= 4:  -1.3158e-16 +0.0000e-01.
```

```
n= 5:  -2.7019e-02 +2.7019e-02.
n= 6:  -3.7526e-02 +0.0000e-01.
n= 7:  -1.3785e-02 -1.3785e-02.
n= 8:  +1.6448e-16 +0.0000e-01.
n= 9:  -8.3392e-03 +8.3392e-03.
n=10:  -1.3509e-02 +0.0000e-01.
```

```
> f_reihe:=
        a[0] +
        add(a[i]*cos(i*2*Pi/T*t), i=1..N) +
        add(b[i]*sin(i*2*Pi/T*t), i=1..N):
```

```
> plot([f, f_reihe], t=0..T,
                      color=[black, red]);
```



Hinweise

Bei der numerischen Berechnung der Fourier-Koeffizienten dürfen in der Funktion keine Parameter enthalten sein.

Bei obiger Rechnung wird die inerte Form des **int**-Befehls verwendet, d.h. das Integral wird zunächst nicht ausgewertet, sondern mit **evalf**(**Int**(..)) wird ein numerisches Integrationsverfahren zur Berechnung des bestimmten Integrals herangezogen. Diese Formulierung ist bei der numerischen Rechnung im Allgemeinen schneller, da dann keine Stammfunktionen berechnet werden, um diese dann an den Integrationsgrenzen auszuwerten.

N spezifiziert die Ordnung der Partialsumme.

Durch die numerische Berechnung der Integrale können Koeffizienten, die analytisch zwar Null sind, nun Werte in der Größenordnung 10^{-9} und kleiner bekommen. Durch die Angabe **Digits**:=*n* wird die Genauigkeit der Rechnung auf *n* Stellen erhöht. Standardmäßig wird mit 10 Stellen gerechnet.

Siehe auch

`int`, **`piecewise`**, **`for-Schleife`**, **`add`**, **`printf`**, **`plot`**, **`Digits`**; → Fourier-Reihen (analytisch) → FFT.

16.3 Komplexe Fourier-Reihe und Amplitudenspektrum

<table>
<tr><td></td><td>worksheet</td></tr>
<tr><td>Problem</td><td>

Gegeben ist eine *T*-periodische Funktion f(*t*). Gesucht sind die komplexen Fourier-Koeffizienten c_n :

$$c_n = \frac{1}{T}\int_0^T f(t)\, e^{(-n\, w_0\, t)}\, dt \qquad \text{mit} \quad w_0 = \frac{2\pi}{T}$$

sowie die Darstellung des Amplitudenspektrums.

</td></tr>
<tr><td>Befehl</td><td>Maple-Befehlsfolge</td></tr>
<tr><td>Parameter</td><td>-</td></tr>
<tr><td>Beispiel</td><td>

f(*t*) = Zweiwegegleichrichter (*T*=1):




```
> w0:=2*Pi/T:
> f1:=i0*sin(w0*t):   #erstes Intervall 0<=t<=T/2
> f2:=-i0*sin(w0*t):  #zweites Intervall T/2<=t<=T

> Integral:=  1/T*Int(f1* exp(-I*n*w0*t),t=0..T/2)
>           + 1/T*Int(f2*exp(-I*n*w0*t),t=T/2..T):
> c[n]:=normal(value(Integral));
```

$$c_n := -\frac{1}{2}\,\frac{i0\,(2\,e^{(-I\,n\,\pi)} + 1 + e^{(-2\,I\,n\,\pi)})}{\pi\,(n^2 - 1)}$$

Man erkennt, dass bei den Koeffizienten c_n durch den Term n^2-1 dividiert wird. Daher sind die Koeffizienten für *n*=1 und *n*=-1 separat zu berechnen:

```
> c[1]:=limit(c[n], n=1);
```

$$c_1 := 0$$

</td></tr>
</table>

	> `c[-1]:=limit(c[n], n=-1);` $c_{-1} := 0$ Das Amplitudenspektrum ist bis auf den Faktor 2 der Betrag der komplexen Fourier-Koeffizienten > `i0:=1:` > `l := [seq([ [k,0], [k, abs(limit(c[n],n=k))]], k=-15..15)]:` > `plot(l, x=-12..12,color=black, labels=[n,``], thickness=3, title=Amplitudenspektrum);` 
Hinweise	Bei der analytischen Berechnung der Fourier-Koeffizienten dürfen in der Funktion Parameter enthalten sein. Damit das Integral aber berechnet wird, sollte auf eine Definition der Funktion über den **piecewise**-Befehl verzichtet werden. Stattdessen sollten die Integrale geeignet aufgespaltet und die Funktionsvorschrift direkt in die Integrale eingesetzt werden.
Siehe auch	**`int`**, **`value`**, **`normal`**, **`limit`**, **`plot`**, **`seq`**; → FFT → Fourier-Reihen (analytisch).

16.4 FFT

Fourier-Transform — worksheet

Problem

Gegeben ist eine T-periodische Funktion f(t) in Form von N Messdaten $f(t_0)$, $f(t_1)$, ..., $f(t_{N-1})$. Gesucht sind Näherungen für die Fourier-Koeffizienten c_m

$$c_m \approx \frac{1}{N} \sum_{n=0}^{N-1} f(t_n)\, e^{\left(-\frac{i\, n\, m\, 2\pi}{N}\right)}$$

sowie die Fourier-Reihe

$$f(t_n) = \frac{1}{T} \sum_{m=0}^{N-1} c_m\, e^{\left(\frac{i\, n\, m\, 2\pi}{N}\right)}$$

Befehl

`FourierTransform` (x, y);

Parameter

x: Vektor der diskreten Realteile der Funktion
y: Vektor der diskreten Imaginärteile der Funktion

Beispiel

Gesucht sind die Amplituden der Frequenzen, die in

$$\sin(2\pi\, 1.0\, t) + \sin(2\pi\, 1.2\, t) + \sin(2\pi\, 2.3\, t) + \sin(2\pi\, 2.7\, t)$$

enthalten sind.

```
> f:=t->evalf(sin(2*Pi*1.0*t) + sin(2*Pi*1.2*t) +
           sin(2*Pi*2.3*t) + sin(2*Pi*2.7*t)):
> plot(f(t),t=0..15,color=black, thickness=2);
```



Abtastung der Funktion mit dem **seq**-Operator:

```
> m:=8: N:=2^m:
> T:=20.1: dt:=evalf(T/N):
> fd := Vector( [seq( f((i-1)*dt), i=1..N)] ):
> imd:= Vector( [seq(0, i=1..N)] ):
```

<table>
<tr>
<td></td>
<td>

Berechnung der FFT

```
> with(DiscreteTransforms):
> Xt,Yt := FourierTransform(fd,imd):
> print(seq(Xt[i],i=1..N));
       print(seq(Yt[i],i=1..N));[5]
```

Graphische Darstellung des Spektrums

```
> plot_data:= seq( [(i-1)*2*Pi/T,
               2*sqrt((Xt[i]^2+ Yt[i]^2)/N)],
                                    i=1..N/2):
> plot([plot_data], color=red, labels=[`w`,``]);
```



</td>
</tr>
<tr>
<td>Hinweise</td>
<td>

FourierTransform aus dem neuen **DiscreteTransforms**-Package ersetzt den alten **FFT**-Befehl zur Berechnung der diskreten Fourier-Transformation. Die neue Variante ist nicht nur wesentlich schneller, sondern sie ist auch dann anwendbar, wenn die Anzahl der diskreten Werte keine Zweierpotenz ist. Die Eingangsgrößen werden durch die Prozedur nicht mehr wie bisher überschrieben. Bei der Anwendung der neuen Prozedur ist zu beachten, dass als Normierungsfaktor $1/\sqrt{N}$ verwendet wird.
Der Real- bzw. Imaginärteil der transformierten Daten müssen für N>4 explizit mit dem **print**-Befehl ausgegeben werden. Mit dem **plot**-Befehl wird das Spektrum graphisch dargestellt.
Verwendet man die Heavisidefunktion (=Sprungfunktion), so muss diese an der Sprungstelle t=0 z.B. durch Heaviside(0.):=0 definiert werden.

</td>
</tr>
<tr>
<td>Siehe auch</td>
<td>

`Vector`, **`seq`**, **`plot`**, **`Heaviside`**;
→ Fourier-Reihen (analytisch)
→ Komplexe Fourier-Reihe und Amplitudenspektrum.

</td>
</tr>
</table>

[5] Auf die Ausgabe der Daten wird aufgrund von Platzgründen verzichtet.